
1 Aufgabenstellung 
In diesem Beitrag wird demonstriert, wie mit einer VBA-Prozedur die Feiertage in einem 

ausgewählten deutschen Bundesland für ein vorgegebenes Kalenderjahr erzeugt und automatisch in 

den OUTLOOK-Standardkalender eingefügt werden können. Auch das automatische Entfernen der 

erstellten Kalendereinträge wird vorgeführt. 

Zuvor wird dargestellt, warum das Erstellen von Kalendereinträgen mit der eingebauten OUTLOOK-

Kalenderoption „Feiertage in Kalender eintragen“ in der föderativ gegliederten Bundesrepublik 

Deutschland nur bedingt geeignet ist. 

Die gesetzlichen Feiertage sind in den 16 deutschen Bundesländern unterschiedlich geregelt (s. 

Abbildung 1). Welche Tage gesetzliche Feiertage sind, richtet sich nach den Feiertagsgesetzen des 

jeweiligen Bundeslandes. Sie besitzen nicht nur arbeitsrechtliche Bedeutung, sondern spielen auch 

für die Personaleinsatzplanung im Allgemeinen eine wichtige Rolle, beispielweise für die 

Terminplanung von Freiberuflern. 

 

Abbildung 1: Gesetzliche Feiertage in 16 deutschen Bundesländern (Quelle [1]) 

2 Feiertage mit eingebauter OUTLOOK-Funktion in den 

Standardkalender eintragen 
OUTLOOK 2010 besitzt eine eingebaute Funktion, um Feiertage automatisch in den standardmäßigen 

Kalender einzutragen1 (s. Abbildung 2). 

 

Abbildung 2: OUTLOOK-Kalenderoption: Feiertage in Kalender eintragen 

                                                             
1 Navigation: Datei > Optionen > Kalender > Kalenderoptionen > Feiertage hinzufügen 

Die Schaltfläche "Feiertage hinzufügen" (s. Abbildung 2) erzeugt insgesamt 276 Kalendereinträge, 
davon für 12 Jahre 8 bewegliche Feiertage (s. Abbildung 3, gelb markiert) und für 20 Jahre andere 
Feiertage. 

 

Abbildung 3: 276 erzeugte Kalendereinträge bei Anwendung der OUTLOOK-Standardfunktion 

Der Reformationstag ist aber nur in 9 von 16 Bundesländern ein gesetzlicher Feiertag (s. Abbildung 1). 

Fronleichnam ist nur in 6 Bundesländern ausnahmslos ein gesetzlicher Feiertag, in Sachsen und 

Thüringen aber nur teilweise (s. Abbildung 4). 

Schalter Bundesland Name des Feiertags Eingrenzung … 

bolKatholisch 
Bayern (BY) Mariä Himmelfahrt bei überwiegend katholischer 

Bevölkerung Thüringen (TH) Fronleichnam 

bolFronleichnam Sachsen (SN) Fronleichnam 
für best. Gemeinden in den 
Kreisen Bautzen u. Westlausitz 

bolFriedensfest Bayern (BY) Friedensfest Stadtgebiet von Augsburg 

Abbildung 4: Boolesche Schalter zur Bestimmung best. gesetzlicher Feiertage durch den Benutzer 

Feiertage wie das Augsburger Friedenfest in Bayern, der Weltkindertag in Thüringen sowie der 

Internationale Frauentag in Berlin fehlen in der Auflistung (s. Abbildung 3). Kurz, bei Nutzung der 

eingebauten Funktion "Feiertage hinzufügen" durch den Benutzer muss der Standardkalender in 

OUTLOOK händisch angepasst werden. Das ist bei 276 automatisch erzeugten Kalendereinträgen für 

12 bzw. 20 Jahre eine ziemliche Herausforderung. 

3 Benutzerformular erstellen 
Aufgrund der Unzulänglichkeiten der eingebauten Funktion "Feiertage hinzufügen" für die praktische 

Terminplanung biete es sich an, eine maßgeschneiderte Lösung für das Hinzufügen von Feiertagen in 

den OUTLOOK-Standardkalender zu erstellen. Basis dafür ist ein benutzerdefiniertes Formular, das in 

OUTLOOK erstellt wird. Die Gestaltung dieses Formulars zeigen Abbildung 5 und Abbildung 6. 

Abbildung 5: Benutzerformular mit Registerblatt 0 des Multiseiten-Steuerelements 

 

Abbildung 6: Benutzerformular mit Registerblatt 1 des Multiseiten-Steuerelements 

Weitere Einzelheiten des benuterdefinierten Formulars sind in den folgenden Tabellen 1, 2 und 3 

enthalten: 

Steuerelement Typ Beschriftung 

lblTitle Bezeichnungsfeld OUTLOOK-Kalendereinträge einfügen oder entfernen 

multipage1 Multiseiten frmMultiSeiten 

cmdOK Befehlsschaltfläche OK 

cmdCancel Befehlsschaltfläche Abbruch 

Tabelle 1: Die vier Steuerelemente des benutzerdefinierten Formulars 

Registerblatt Index Beschriftung 

pgInsert 0 einfügen 

pgRemove 1 entfernen 

Tabelle 2: Die beiden Registerblätter des Multiseiten-Steuerelements 

Steuerelement Typ Beschriftung 

Steuerelemente im Registerblatt 'pgInsert' 

lblJahr Bezeichnungsfeld Feiertage im Jahr 

txtJahr Textfeld  

lblSelection Bezeichnungsfeld Bitte eine deutsches Bundesland auswählen 

cboLand Kombinationsfeld  

fraSpecial Rahmen Regelungen für die Länder [BY, SN, TH] 

chkKatholisch Kontrollkästchen BY: Überwiegend katholische Bevölkerung? 

chkAugsburg Kontrollkästchen BY: Augsburger Friedensfest? 

chkFronleichnam Kontrollkästchen SN u. TH: Fronleichnam? 

Steuerelemente im Registerblatt 'pgRemove' 

chkRemove Kontrollkästchen Wirklich alle Feiertage entfernen? 

Tabelle 3: Die neun Steuerelemente in den Registerblättern des Multiseiten-Steuerelements 

Der VBA-Quellcode des benutzerdefinierten Formulars ist in Listing 1 enthalten. Er enthält fünf 

Subroutinen mit folgenden Aufgaben (s. Tabelle 4). 

Name Aufgabe 

UserForm_Initialize() Setzt Anfangswerte für das Benutzerformular 

cboLand_Change() Reagiert auf eine Änderung des Kombinationsfelds 'cboLand' 

UserForm_QueryClose(…) Verhindert das Schließen des Benutzerformulars mit dem roten Kreuz 

cmdOK_Click() Ruft eine zugeordnete Prozedur auf und führt sie aus. 

cmdCancel_Click() Beendet das Benutzerformular 

Tabelle 4: Subroutinen des Benutzerformulars 

4 Feiertage in den Standardkalender einfügen 
Für das ausgewählte Jahr (2019) und das ausgewählte deutsche Bundesland (NW – Nordrhein-

Westfalen) (s. Abbildung 5) erstellt die Prozedur InsertCalendarItems die entsprechenden Einträge in 

den OUTLOOK-Standardkalender. Diese Prozedur benötigt 5 Argumente: 

Name des Arguments Inhalt des Arguments 

intYr (aktuelles) Kalenderjahr 

strLand Langname eines dt. Bundeslandes 

bolFriedensfest boolescher Schalter für Augsburger Friedensfest 

bolFonleichnam boolescher Schalter für Fronleichnam 

bolKatholisch boolescher Schalter für überwiegend kath. Bevölkerung 

Tabelle 5: Argumente der Prozedur InsertCalendarItems 

Die Prozedur InsertCalendarItems (s. Listing 2) benötigt zwei Subroutinen: 

Name der Subroutine Aufgabe der Subroutine 

SetRecurringHolidaysFromYear Wiederkehrende Feiertage ab Jahr intYr für das Land strKuerzel in 
den Standardkalender eintragen. 

 Führt für jeden festen Feiertag die Routine 
CreateRecurringAllDayEvent aus. 

 Ruft dort jeweils die Funktion AllDayEventExists auf. 

 Erstellt dann ggf. einen wiederkehrenden Termin. 

SetMoveableHolidaysForThisYear Bewegliche Feiertage des Jahres intYr für das Land strKuerzel in 
den Standardkalender eintragen: 

 Bestimmt bewegliche Feiertage mit den beiden 
Funktionen EasterSunday und BussUndBettag. 

 Führt für jeden beweglichen Feiertag die Subroutine 
CreateSingleAllDayEvent aus. 

 Ruft dort jeweils die Funktion AllDayEventExists auf. 

 Erstellt dann ggf. einen einmaligen Termin. 

Tabelle 6: Subroutinen der Prozedur InsertCalendarItems 

Die Prozedur SetMoveableHolidaysForThisYear (s. Tabelle 6, 1. Spalte, 3. Zeile) benötigt drei 

Funktionen (s. Tabelle 7): 

Name der Funktion Aufgabe der Funktion 

EasterSunday Ostersonntag eines best. Jahres ermitteln. 

BussUndBettag Das Datum des Buß- und Bettags eines best. Jahres ermitteln. 

AllDayEventExists Gibt ‘True’ zurück, wenn ein Kalendereintrag am angegebenen 
Datum und Betreff bereits vorhanden ist, sonst ‘False’. 

Tabelle 7: Benötigte Funktionen für die Prozedur SetMoveableHolidaysForThisYear 

Bei Anwendung der Prozedur InsertCalendarItems mit den beiden Argumenten (vgl. Abbildung 5) 

 NW (für Nordrhein-Westfalen) 

 2019  (für das Jahr 2019 

wird beispielsweise diese Erfolgsmeldung ausgegeben (s. Abbildung 7): 

Abbildung 7: Erfolgsmeldung nach Erstellung der Feiertage 2019 in 'Nordrhein-Westfalen' 

Unter der Kategorie „Feiertage“ sind die 13 relevanten Feiertage im OUTLOOK-Standardkalender zu 

erkennen (s. Abbildung 8) 2. Davon entfallen 2 Feiertage (Ostersonntag u. Pfingstsonntag) auf ein 

Wochenende. Die anderen 11 Feiertage sind sog. Wochenfeieretage, weil sie nicht auf ein 

Wochenende (Sa/So) fallen. 

 

Abbildung 8: 13 Feiertage im Bundesland 'Nordrhein-Westfalen' im Jahr 2019 

4.1 Duplikate verhindern 
Die Subroutine AllDayEventExists (s. Listing 2) verhindert doppelte Kalendereinträge, wenn die 

Prozedur InsertCalendarItems versehentlich mit den zuvor gewählten Argumenten erneut ausgeführt 

wird, ohne dass vorher alle Kalendereinträge der Kategorie Feiertage entfernt wurden. In diesem Fall 

wird folgende Fehlermeldung ausgegeben: 

                                                             
2 11 bundesweite/deutschlandweite Feiertage sowie Fronleichnam (Do) und Allerheiligen (Fr). 

Abbildung 9: Die Routine AllDayEventExists verhindert doppelte Kalendereinträge  

 

4.2 Nur Wochenfeiertage in den Standardkalender einfügen 
Um dafür zu sorgen, dass nur sog. Wochenfeiertage in den OUTLOK-Standardkalender eingefügt 

werden, kann die Funktion IsWeekendAppointment (s. Listing 4) mit dem Argument dtmStart (d. h. 

Startdatum) eines einmaligen oder wiederkehrenden Termins in den Ablauf der Prozedur 

InsertCalendarItems eingebaut werden3. 

5 Feiertage aus dem Standardkalender entfernen 
Das Entfernen aller Kalendereinträge der Kategorie Feiertage aus dem OUTLOOK-Standardkalender 

erfolgt via Benutzerformular (vgl. Abbildung 6), indem die Prozedur DeleteCalendarItemsByCategory 

(s. Listing 3) ausgeführt wird. Abbildung 10 beinhaltet die entsprechende Erfolgsmeldung. 

 

Abbildung 10: Erfolgsmeldung nach dem Entfernen aller Feiertage aus dem Standardkalender 

 

                                                             
3 Wochenfeiertage fallen auf einen Montag bis inkl. Samstag. Die Funktion IsWeekendAppointment betrachtet 
aber Feiertage, die auf einen Samstag fallen, nicht als Wochenfeiertag. Es wird eine 5-Tage-Woche 
angenommen. 

6 VBA-Quellcode 

6.1 Benutzerformular erstellen 

Option Explicit 

' Globale Konstante und Variable definieren 

Const conCategory As String = "Feiertage" ' OUTLOOK-Kategorie 

Dim intCreated    As Integer              ' Zähler f. erstellte Kalendereinträge 

 

Private Sub UserForm_Initialize() 

   Dim arrState()  As String 

   ' Mit der Split-Funktion ein 0-basiertes eindimensionales Array mit 16 Elementen erstellen. 

    arrState = Split("BW Baden-Würtemberg|BY Bayern|BE Berlin|BB Brandenburg|" & _ 

                   "HB Bremen|HH Hamburg|HE Hessen|MV Meckenburg-Vorpommerm|" & _ 

                   "NI Niedersachsen|NW Nordrhein-Westfalen|RP Rheinland-Pfalz|" & _ 

                   "SL Saarland|SN Sachsen|ST Sachsen-Anhalt|SH Schleswig-Holsten|TH Thüringen","|") 

 

   With MultiPage1                           ' Multiseiten-Steuerelement 

      .Value = 0                             ' erste Registerseite anzeigen 

      .Style = fmTabStyleTabs                ' Standard-Stil 

      .TabOrientation = fmTabOrientationTop  ' Register oben anordnen 

      .TabFixedWidth = 0                     ' automatische Breite 

      .TabFixedHeight = 0                    ' automatische Höhe 

   End With 

   With Me 

      .cboLand.List = arrState               ' Datenquelle des Kombinationsfelds bestimmen 

      .cboLand.SetFocus                      ' Kombinationsfeld fokussieren 

      .cboLand.Value = arrState(9)           ' Kombinationsfeld mitAnfangswert (9. Array-Element) belegen 

      .txtYear = Year(Date)                  ' aktuelles Kalenderjahr bestimmen 

      .chkKatholisch = .chkAugsburg = .chkFronleichnam = False 

      .chkKatholisch.Enabled = False 

      .chkAugsburg.Enabled = False 

      .chkFronleichnam.Enabled = False 

   End With 

   intCreated = 0 

End Sub 

 

Private Sub cboLand_Change() 

   Dim intResponse,intStyle  As Integer 

   Dim strMsg,strTitle       As String 

   strTitle = "Gesetzlicher Feiertag?" 

   intStyle = vbYesNo + vbQuestion + vbDefaultButton2 

   With Me 

      .chkKatholisch = False 

      .chkAugsburg = False 

      .chkFronleichnam = False 

   End With 

   Select Case Left(Me.cboLand.Value,2) 

      Case "BY" 

         strMsg = "BY: Ist Augsburger Friedenfest ein Feiertag?" 

         intResponse = MsgBox(strMsg,intStyle,strTitle) 

         If intResponse = vbYes Then 

            With Me.chkAugsburg 

               .Enabled = True 

               .Value = True 

               .Enabled = False 

            End With 

         End If 

         strMsg = "BY: Mariä Himmelfahrt ein Feiertag?" 

         intResponse = MsgBox(strMsg,intStyle,strTitle) 

         If intResponse = vbYes Then 

            With Me.chkKatholisch 

               .Enabled = True 

               .Value = True 

               .Enabled = False 

            End With 

         End If 

      Case "SN","TH" 

         strMsg = "SN: Ist Fronleichnam ein Feiertag?" 

         intResponse = MsgBox(strMsg,intStyle,strTitle) 

         If intResponse = vbYes Then 

            With Me.chkFronleichnam 

               .Enabled = True 

               .Value = True 

               .Enabled = False 

            End With 

         End If 

   End Select 

   If Me.cboLand.ListIndex = -1 Then 

      Me.cboLand.SetFocus 

   End If 

End Sub 

 

Private Sub UserForm_QueryClose(Cancel As Integer,CloseMode As Integer) 

   If CloseMode = vbFormControlMenu Then 

      MsgBox "Bitte das Formular mit der Schaltfläche 'Abbruch' schließen!" 

      Cancel = True 

   End If 

End Sub 

 

Private Sub cmdOK_Click() 

   On Error GoTo Exit_Point 

   Me.Hide 

   If Me.MultiPage1.Value = 0 Then 

      With Me 

         ' Kalendereinträge erstellen (mit 5 Argumenten) 

         Call InsertCalendarItems(.txtYear, .cboLand, _ 

              .chkAugsburg, .chkKatholisch, .chkFronleichnam) 

      End With 

   End If 

   If Me.MultiPage1.Value = 1 Then 

      If Me.chkRemove = True Then 

         ' Kalendereinträge entfernen (mit OUTLOOK-Kategorie) 

         Call DeleteCalendarItemsByCategory(conCategory) 

      End If 

   End If 

Exit_Point: 

   Unload Me 

End Sub 

 

Private Sub cmdCancel_Click() 

   Me.Hide 

   Unload Me 

End Sub 

Listing 1: Benutzerformular erstellen 

6.2 Kalendereinträge erstellen 

Private Sub InsertCalendarItems(intYr As Integer,strLand As String,_ 

        bolFriedensfest As Boolean,bolKatholisch As Boolean,bolFronleichnam As Boolean) 

   ' Aufgabe: Dieses VBA-Skript fügt dem Outlook-Standardkalender die Feiertage eines best. 

      Jahres hinzu. 

   ' Argumente: 

   '     intYr                aktuelles Kalenderjahr 

   '     strLand              Langname für dt. Bundesland 

   '     bolFriedensfest      boolescher Schalter für Augburger Friendsfest in BY 

   '     bolFonleichnam       boolescher Schalter für Fronleichnam 

   '     bolKatholisch        boolescher Schalter für überwiegend kath. Bevölkerung 

   Dim strKuerzel As String * 2 

   On Error GoTo Error_Point 

   ' Deutsches Bundesland ermitteln 

   strKuerzel = UCase(Left(strLand,2))    ' Kurzname f. Bundesland 

   strLand = Mid(strLand,4)               ' Langname f. Bundesland 

   intCreated = 0                         ' Anfangswert für Zähler 

   ' Feste Feiertage des Jahres intYr im Land strKuerzel in den Standardkalender eintragen. 

   Call SetRecurringHolidaysFromYear(intYr,strKuerzel,bolFriedensfest,bolKatholisch) 

   ' Bewegliche Feiertage d. Jahres intYr im Land strKuerzel in d. Standardkalender eintragen. 

   Call SetMoveableHolidaysForThisYear(intYr,strKuerzel,bolFronleichnam) 

   MsgBox strLand & ": " & vbCrLf & CStr(intCreated) & " Feiertage in d. Standarkalender 

eingefügt!",vbInformation,"Feiertage erstellen" 

   Me.Show ' Benutzerformular anzeigen 

Exit_Point: 

   Exit Sub 

Error_Point: 

   MsgBox "Laufzeitfehler (" & Err.Number & ") " & Err.Description,vbCritical,"Feiertage" 

   Resume Exit_Point 

End Sub 

 

Private Function EasterSunday(ByVal lngYear As Long) As Date 

   ' Aufgabe: Ostersonntag eines best. Jahres ermitteln. 

   Dim lngDiff  As Long 

   Dim lngDelta As Long 

   ' Wenn kein Jahr übergeben wird, das aktuelle Jahr verwenden. 

   If lngYear = 0 Then lngYear = Year(Now) 

   lngDiff = (((255 - 11 * (lngYear Mod 19)) - 21) Mod 30) + 21 

   lngDelta = lngDiff + IIf(lngDiff > 48,1,0) + 6 - _ 

     ((lngYear + Int(lngYear / 4) + lngDiff + IIf(lngDiff > 48,1,0) + 1) Mod 7) 

   ' Osterdatum zurückgeben 

   EasterSunday = DateAdd("d",lngDelta,DateSerial(lngYear,3,1)) 

End Function 

 

Private Function BussUndBettag(intY As Integer) As Date 

   ' Aufgabe: Buß- und Bettag eines Jahres berechnen 

   BussUndBettag = DateSerial(intY,11,23) 

   BussUndBettag = BussUndBettag - Weekday(BussUndBettag,vbThursday) 

End Function 

 

Private Function AllDayEventExists(ByVal dtmDate As Date,ByVal strSubject As String) As 

Boolean 

   ' Aufgabe: Gibt True zurück, wenn der Kalendereintrag am angegebenen Datum und Betreff 

   vorhanden ist. 

   Dim blnExists  As Boolean 

   Dim objFolder  As Outlook.Folder 

   Dim objItems   As Outlook.Items 

   Dim strSearch  As String 

   blnExists = False 

   strSearch = "[Start] = '" & FormatDateTime(dtmDate,vbShortDate) & " 0:00' And [Subject] = 

'" & strSubject & "'" 

   Set objItems = 

Outlook.Application.GetNamespace("MAPI").GetDefaultFolder(olFolderCalendar).Items 

   objItems.Sort "[Start]" 

   objItems.IncludeRecurrences = True 

   Set objItems = objItems.Restrict(strSearch) 

   If Not objItems Is Nothing Then 

      If objItems.Count > 0 Then 

         Dim objItem As Object 

         Set objItem = objItems.GetFirst() 

         If Not objItem Is Nothing Then 

            blnExists = True 

         End If 

      End If 

   End If 

   Set objItem = Nothing 

   AllDayEventExists = blnExists 

End Function 

 

Private Sub CreateSingleAllDayEvent(ByVal dtmDate As Date,ByVal strSubject As String) 

   ' Aufgabe: Erstellt ein ganztägiges Ereignis am angegebenen Datum und Betreff, 

   ' wenn dieses ganztägige Ereignis noch nicht vorhanden ist. 

   If Not AllDayEventExists(dtmDate,strSubject) Then 

      Dim olAppt As Outlook.AppointmentItem 

      Set olAppt = Outlook.Application.CreateItem(olAppointmentItem) 

      With olAppt 

         .Start = dtmDate 

         .Subject = strSubject 

         .AllDayEvent = True 

         .ReminderSet = False 

         .Categories = conCategory 

         .Save          

         If .BusyStatus <> olBusy Then 

            .BusyStatus = olBusy    ' ... als "gebucht" anzeigen 

            .Save 

         End If 

      End With 

      intCreated = intCreated + 1 

      Set olAppt = Nothing 

   End If 

End Sub 

 

Private Sub CreateRecurringAllDayEvent(ByVal dtmDate As Date,ByVal strSubject As String) 

   ' Aufgabe: Erstellt ein ganztägiges Ereignis zum angegebenen Datum (dtmDate) 

   ' mit dem angegebenen Betreff (strSubject) und legt ein jährliche wiederkehrendes 

   ' Serienmuster an, falls es noch nicht vorhanden ist. 

   If Not AllDayEventExists(dtmDate, strSubject) Then 

      Dim olAppt     As Outlook.AppointmentItem 

      Dim olPattern  As Outlook.RecurrencePattern 

      Set olAppt = Outlook.Application.CreateItem(olAppointmentItem) 

      With olAppt 

         .Start = dtmDate 

         .Subject = strSubject 

         .AllDayEvent = True 

         .ReminderSet = False 

         .Categories = conCategory 

         If .BusyStatus <> olBusy Then 

            .BusyStatus = olBusy    ' ... als "gebucht" anzeigen 

         End If 

         Set olPattern = .GetRecurrencePattern() 

         With olPattern ' Serienmuster: jährlich 

            .RecurrenceType = olRecursYearly 

            .DayOfMonth = Day(dtmDate) 

            .MonthOfYear = Month(dtmDate) 

            .PatternStartDate = olAppt.Start 

            .NoEndDate = True 

         End With 

         .Save 

      End With 

      intCreated = intCreated + 1 

      Set olPattern = Nothing 

      Set olAppt = Nothing 

   End If 

End Sub 

 

Private Sub SetMoveableHolidaysForThisYear(ByVal intY As Integer,strKuerzel As String, 

bolFronleichnam As Boolean) 

   ' Aufgabe: Bewegliche Feiertage eines Jahres bestimmen. 

   Dim dtmEaster,dtmBusstag As Date 

   dtmEaster = EasterSunday(intY) 

   Call CreateSingleAllDayEvent(DateAdd("d",-2,dtmEaster),"Karfreitag") 

   Call CreateSingleAllDayEvent(dtmEaster,"Ostersonntag") 

   Call CreateSingleAllDayEvent(DateAdd("d",1,dtmEaster),"Ostermontag") 

   Call CreateSingleAllDayEvent(DateAdd("d",39,dtmEaster),"Chr. Himmelfahrt") 

   Call CreateSingleAllDayEvent(DateAdd("d",49,dtmEaster),"Pfingstsonntag") 

   Call CreateSingleAllDayEvent(DateAdd("d",50,dtmEaster),"Pfingstmontag") 

   If InStr(1,"BW BY HE NW RP SL",strKuerzel) > 0 Then 

      Call CreateSingleAllDayEvent(DateAdd("d",60,dtmEaster),"Fronleichnam") 

   Else 

      If InStr(1,"SN TH",strKuerzel) > 0 And bolFronleichnam Then 

         Call CreateSingleAllDayEvent(DateAdd("d",60,dtmEaster),"Fronleichnam") 

      End If 

   End If 

   If strKuerzel = "SN" Then ' Sachsen 

      dtmBusstag = BussUndBettag(intY) 

      Call CreateSingleAllDayEvent(dtmBusstag,"Buß- u. Bettag") 

   End If 

End Sub 

 

Private Sub SetRecurringHolidaysFromYear(ByVal intY As Integer,ByVal strKuerzel As String,_ 

   bolKatholisch As Boolean,bolFriedensfest As Boolean) 

   ' Aufgabe: Feste Feiertage eines Jahres setzen. 

   Call CreateRecurringAllDayEvent(DateSerial(intY,1,1),"Neujahr") 

    

   If InStr(1,"BW BY ST",strKuerzel) > 0 Then 

      Call CreateRecurringAllDayEvent(DateSerial(intY,1,6),"Hl. 3 Könige") 

   End If 

    

   If strKuerzel = "BE" Then 

      Call CreateRecurringAllDayEvent(DateSerial(intY,3,8),"Int. Frauentag") 

   End If 

    

   Call CreateRecurringAllDayEvent(DateSerial(intY,5,1),"Tag d. Arbeit") 

    

   If strKuerzel = "BY" And bolFriedensfest Then 

      Call CreateRecurringAllDayEvent(DateSerial(intY,8,8),"Augsb. Frieden") 

   End If 

    

   If strKuerzel = "SL" Or (strKuerzel = "BY" And bolKatholisch) Then 

      Call CreateRecurringAllDayEvent(DateSerial(intY,8,15),"Mariä Himmelfahrt") 

   End If 

    

   If strKuerzel = "TH" Then 

      Call CreateRecurringAllDayEvent(DateSerial(intY,9,20),"Weltkindertag") 

   End If 

    

   Call CreateRecurringAllDayEvent(DateSerial(intY,10,3),"Tag der Deutschen Einheit") 

   If InStr(1,"BB HB HH MV NI SN ST SH TH",strKuerzel) > 0 Then 

      Call CreateRecurringAllDayEvent(DateSerial(intY,10,31),"Reformationstag") 

   End If 

    

   If InStr(1,"BW BY NW RP SL",strKuerzel) > 0 Then 

      Call CreateRecurringAllDayEvent(DateSerial(intY,11,1),"Allerheiligen") 

   End If 

    

   Call CreateRecurringAllDayEvent(DateSerial(intY,12,25),"1. Weihnachtstag") 

    

   Call CreateRecurringAllDayEvent(DateSerial(intY,12,26),"2. Weihnachtstag") 

End Sub 

Listing 2: Kalendereinträge erstellen 

6.3 Kalendereinträge entfernen 

Private Sub DeleteCalendarItemsByCategory(conCategory As String) 

   ' Aufgabe: Alle Kalendereinträge einer bestimmten Kategorie (conCategory) entfernen. 

   Dim olFolder   As Outlook.MAPIFolder      ' Verzeichnis 

   Dim olItems    As Outlook.Items           ' Kalendereinträge 

   Dim olAppt     As Outlook.AppointmentItem ' Termin 

   Dim intLoop    As Integer                 ' Schleifenzähler 

   Dim intCnt     As Integer                 ' Löschzähler 

   Set olFolder = Outlook.Application.GetNamespace("MAPI").GetDefaultFolder(olFolderCalendar) 

   Set olItems = olFolder.Items 

   For intLoop = olItems.Count To 1 Step -1 

      Set olAppt = olItems.item(intLoop) 

      With olAppt 

         If .Categories = conCategory Then 

            .Delete 

            intCnt = intCnt + 1 

         End If 

      End With 

   Next intLoop 

   MsgBox "Kategorie: " & conCategory & vbCrLf & Str(intCnt) & " Kalendereinträge entfernt", 

vbOKOnly + vbInformation, "Globales Löschen mit Kategorie" 

   Set olItems = Nothing 

   Set olFolder = Nothing 

   Me.Show ' Benutzerformular anzeigen 

   ' Outlook.Application.Quit ' Anwendung beenden 

End Sub 

Listing 3: Kalendereinträge entfernen 

6.4 Termine am Wochenende erkennen 

Private Function IsWeekendAppointment(ByVal dtmStart As Date) As Boolean 

   ' Aufgabe: Termin/Ereignis am Wochenende (Sa., So.) erkennen. 

   Select Case Weekday(dtmStart, vbSunday) 

      Case vbSaturday, vbSunday 

         IsWeekendAppointment = True 

      Case Else 

         IsWeekendAppointment = False 

    End Select 

End Function 

Listing 4: Termine am Wochenende erkennen 

